# Author: williamjcm <w.jcm59@gmail.com>
# Contributor: mosra <mosra@centrum.cz>
# Based on the packages/archlinux/PKGBUILD file by mosra <mosra@centrum.cz> and the MSYS2/MinGW PKGBUILD templates

_realname=corrade
pkgbase=mingw-w64-${_realname}
pkgname="${MINGW_PACKAGE_PREFIX}-${_realname}"
pkgver=dev
pkgrel=1
pkgdesc="C++11/C++14 multiplatform utility library"
arch=('any')
url="https://magnum.graphics/corrade/"
license=('MIT')
makedepends=("${MINGW_PACKAGE_PREFIX}-cmake" "${MINGW_PACKAGE_PREFIX}-ninja")
options=(!strip)

_rootdir=$startdir/../../

build() {
    mkdir -p "$_rootdir/build-${CARCH}"
    cd "$_rootdir/build-${CARCH}"

    MSYS2_ARG_CONV_EXCL="-DCMAKE_INSTALL_PREFIX=" \
        ${MINGW_PREFIX}/bin/cmake .. \
            -G'Ninja' \
            -DCMAKE_BUILD_TYPE=Debug \
            -DCMAKE_INSTALL_PREFIX=${MINGW_PREFIX}
    ninja
}
package() {
    cd "$_rootdir/build-${CARCH}"
    DESTDIR="${pkgdir}" ninja install
}
